

Paper 348-2011**Rasch Model and Beyond with PROC GLIMMIX: A Better Choice****Teck Kiang Tan, Luo Wen Su Serena, David Hogan****Centre for Research in Pedagogy and Practice, Nanyang Technological University, Singapore****ABSTRACT**

Rasch model is commonly used in education measurement and testing for evaluating tests and assessments. While there are various SAS® procedures that could carry out item analysis, PROC GLIMMIX provides a relatively simple syntax for Rasch modeling and it takes almost 500 times faster to estimate a regular Rasch model compared to PROC NLMIXED. This paper demonstrates how to estimate standard Rasch model using PROC GLIMMIX, and further illustrates the possibilities of using this procedure, under the multilevel framework, to estimate various Rasch models. The paper ends with Rasch models that incorporate covariates at person and item levels to explain the probability of getting a correct answer for an assessment.

INTRODUCTION

Rasch model (1960) is commonly used in modern education measurement and testing. Unlike classical measurement, it takes into account the difficulty level of items. This property makes it attractive for modeling assessment as items in tests would not be treated as equally important. The Rasch model is based on a probabilistic relationship between an item level of difficulty and a student's ability. The differences between these two measures govern the probability of giving a correct answer to an item. As this model has one parameter, the difficulty parameter, that defines the probability of response, it is called the one-parameter logistic model, abbreviated as 1PL. The literature generally uses SAS® procedures NLMIXED to model the 1PL as it could be specified as a non-linear mixed model with person specified as random and item specified as fixed. However, articles on Rasch have also shown that regular Rasch models could also be formulated as cross-classified binary logistic models (Adams, Wilson, and Wu, 1997; Kamata, 2001; Van den Noortgate, De Boeck, and Meulders, 2003). This reformulation of Rasch model provides an alternative way to build it using PROC GLIMMIX. More importantly, covariates could be easily incorporated into the model by specifying it as a cross-classification multilevel binary logistic model to explain variation in students' performance. This paper uses a mathematics instrument which consists of 28 items to illustrate the various Rasch models using PROC GLIMMIX.

The paper starts with a short introduction to the Rasch model and the requirements on the data input to estimate the Rasch model. The syntax of PROC NLMIXED and PROC GLIMMIX are given and their results are compared to standard IRT software MULTILOG-MG, and R Package eRm (extended Rasch Models). As expected, similar estimated coefficients are obtained. Using a mathematics assessment as an illustration, the disadvantages of using PROC NLMIXED, as compared to PROC GLIMMIX, are highlighted. Next, we introduce four types of Rasch models and illustrate the usefulness of the two of the RASCH models: the Random Item Fixed Person Rasch model, and the Random Item Random Person Rasch model. In particular, under the multilevel framework, the latter Rasch model could incorporate covariates by specifying it as a cross-classified model. This model is helpful for analysis that takes into accounts of factors, at both the item and student levels, to examine their effects on student's performance. The last section concludes the paper and suggests the use of PROC GLIMMIX for Rasch modeling.

Data Source and Sample

The data are obtained from the Core 2 Panel 2 Project conducted by the Centre for Research in Pedagogy and Research, Nanyang Technological University, Singapore during the early year of 2010. 815 students take the instrument comprising of 28 multiple choice questions on mathematics. There are four domains for the mathematics items namely algebra, geometry, statistics, and number. The survey also asks students about their level of confidence in answering these items. For each mathematics item, the students rate their confidence level from 0% to 100% in answering it. The chance of getting a correct answer for an item generally refers to the objective probability whereas the opinion of students expressed on their level of confidence is the subjective probability of getting a correct answer of an item. The paper uses both these sets of variables as examples for illustration of the various Rasch models using PROC GLIMMIX.

Data Specification

The data structure required for both PROC NLMIXED and PROC GLIMMIX to estimate the Rasch model is shown in the Table 1 below. The response variable is a binary outcome of a correct answer coded as one and incorrect answer coded as zero. In order to find out the item difficulty level, we need to specify a list of dummies of all the items. X_{1i} specifies the first dummy variable for item 1, X_{2i} specifies the dummy variable for item 2, and so on. We may also want to compare person abilities for the entire sample. As such, we have to specify dummy variables for person by specifying each person with a dummy code. Similarly, for person dummy variables, Z_{1j} specifies the first person who takes the assessment, Z_{2j} specifies the second person who takes the test, and so on. For instance, the first observation indicates the response of y_{11} which is the first student Z_{1j} who takes the first item X_{1i} . The second response is y_{21} , with the same student dummy variable Z_{1j} coded as one, who takes the second items X_{2i} coded as one. Whether we use the set of item dummy variables or the person dummy variables would depend on the Rasch model we used for modeling.

Table 1 Data Structure for PROC GLIMMIX

Response Variable	Item Dummy Variable Random Person Effects				Person Dummy Variable Random Item Effects			
	X_{1i}	X_{2i}	...	X_{Ii}	Z_{1j}	Z_{2j}	...	Z_{Ij}
y_{11}	1	0	...	0	1	0	...	0
y_{21}	0	1	...	0	1	0	...	0
\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots
y_{I1}	0	0	...	1	1	0
\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots
y_{1J}	1	0	...	0	0	0	0	1
\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots	\vdots
y_{IJ}	0	0	0	1	0	0	0	1

Rasch Model

The basic Rasch model with one parameter specification is as follows:

$$\text{Logit}(\pi_{ij} | \theta_j) = \ln\left(\frac{\pi_{ij}}{1 - \pi_{ij}}\right) = \theta_j - \delta_i \quad Y_{ij} \sim \text{Bernoulli}(\pi_{ij})$$

$$P(X_i = 1) = \frac{\exp(\theta_j - \delta_i)}{1 + \exp(\theta_j - \delta_i)}$$

$i = 1, \dots, I$ item

$j = 1, \dots, J$ person

π_{ij} = probability person j answer item i correctly

The response of person j to an item i is regarded as a function of the person ability (θ_j) and the item difficulty (δ_i). Generally, Rasch model considers person as random and the item as fixed. This is expected as item used for testing are normally fixed in number and students participated in the test are in general a sample of the population. We shall refer to this form of Rasch model as the regular Rasch in this paper to distinguish it from other Rasch models which we will discuss them later. As Rasch model could be considered as a non-linear mixed model, we could use PROC NLMIXED to model it as shown below.

```
Proc NLMixed Data = LongMathSec3;
  Parms c1= -1.14 c2=-2.72 c3=-0.12 c4=-0.15 c5=-0.64
        c6=-0.20 c7=-0.55 c8=-0.18 c9=0.64 c10=-0.41
        c11=-0.75 c12=-0.92 c13=0.21 c14=-0.46 c15=-0.18
        c16=0.67 c17=-0.05 c18=0.03 c19=-0.33 c20=0.18
        c21=-0.77 c22=1.29 c23=0.09 c24=0.30 c25=-0.65
        c26=-0.82 c27=0.82 c28=0.90
        sigma=1;
  eta = theta -
    (c1*Quest1 + c2*Quest2 + c3*Quest3 + c4*Quest4 + c5*Quest5 +
     c6*Quest6 + c7*Quest7 + c8*Quest8 + c9*Quest9 + c10*Quest10 +
     c11*Quest11 + c12*Quest12 + c13*Quest13 + c14*Quest14 + c15*Quest15 +
     c16*Quest16 + c17*Quest17 + c18*Quest18 + c19*Quest19 + c20*Quest20 +
     c21*Quest21 + c22*Quest22 + c23*Quest23 + c24*Quest24 + c25*Quest25 +
     c26*Quest26 + c27*Quest27 + c28*Quest28);
  expeta = exp(eta);
  p = expeta / (1+ expeta);
  Model LongAccuracy ~ Binary (p);
  Random theta ~ Normal(0, sigma*sigma) Subject=ID;
```

Run;

The PROC NLMIXED is a general procedure that is capable of modeling all kinds of non-linear mixed model as long as the mathematical representation of the model is explicitly stated in the procedure. For the above example, the mathematical representations of the regular Rasch model are specified under three statements: eta, expeta, and p. The MODEL statement states that it is a binary model for the dependent variable LongAccuracy that consists of one for a correct answer and zero for a wrong answer. The RANDOM statement states that person ability is random with normal distribution of mean zero and standard deviation as sigma. The SUBJECT statement states that the subject is the student ID.

Rasch Model as Multilevel Logistic Model

Regular Rasch model could be regarded, under the multilevel framework, as a hierarchical two-level binary logistic model (Kamata, 2001; Van den Noortgate et al, 2003). This gives us an alternative way of specifying the model using another procedure GLIMMIX. PROC GLIMMIX is a procedure that estimates generalized mixed model which include the multilevel binary logistic model. The regular Rasch model could be modeled by specifying a two-level logistic model. The first level is the item level and the second level the person level. As shown in Table 1, we use the set of item dummy variables for the regular Rasch by creating a set of these dummy variables each for an item. The estimated coefficients of PROC GLIMMIX after adjusting for the estimated intercept would be the difficulty level of the item. The mathematical representation of the multilevel binary logistic model is given below.

$$\text{Logit}(\pi_{ij}) = \beta_{oj} + \sum_{k=1}^{I-1} \beta_{kj} X_{ki}, \quad Y_{ki} \sim \text{Bernoulli}(\pi_{ij}), X_{ki} = \begin{cases} 1 & \text{if } k = i \\ 0 & \text{otherwise} \end{cases}$$

$$\beta_{oj} = \gamma_{00} + u_{0j}$$

$$\beta_{1j} = \gamma_{10}$$

$$\vdots$$

$$\beta_{(I-1)j} = \gamma_{(I-1)0}$$

$$u_j \sim N(0, \sigma_u^2)$$

SAS® PROCEDURE GLIMMIX

As PROC GLIMMIX is the procedure for analyzing generalized mixed models, SAS® has set the syntax of PROC GLIMMIX close to PROC MIXED. As such, the syntax is relatively simpler than the general procedure of PROC NLMIXED. Unlike PROC NLMIXED, we do not need to specify the mathematical representation explicitly in the procedure. The syntax of PROC GLIMMIX to estimate a regular Rasch model that consists of the 28 mathematics items is shown below. The CLASS statement states that the student ID is a classification variable. The MODEL statement specifies that the dependent variable LongAccuracy is a binary variable with 1 as correct answer (Event='1') and zero as wrong answer. Variables Quest2 to Quest28 are the dummy variables, each representing an item of the test question. As there are 28 items, only 27 dummies are included in the model. We could choose any item as the reference and adjust the estimated coefficients to derive the difficulty parameters for all the items. The RANDOM statement states that the INTERCEPT is random i.e. person is random. The SUBJECT statement states that student ID is the subject. Comparatively, the specification of PROC GLIMMIX is much simpler than the PROC NLMIXED.

```
Proc GLIMMIX Data = LongMathSec3;
  Class ID;
  Model LongAccuracy (Event='1') =
    Quest2-Quest28
    / Dist = Binary Solution;
  Random Intercept / Subject= ID;
Run;
```

Table 2 below shows the estimated difficulty coefficients of PROC NLMIXED, PROC GLIMMIX, MULTILOG-MG, and R package eRm (extended Rasch Model). They give similar results. The correlation coefficients of the four sets of

difficulty estimates are higher than .99. The difficulty index is the percentage of students who obtained the answer correct for the item. This is one of the indices commonly used in classical measurement. The estimated coefficients of PROC NL MIXED, PROC GLIMMIX , MULTILOG-MG, and R Package eRm indicate the difficulty level of the items. For instance, the estimated coefficient under PROC GLIMMIX for items 1 and 2 are -1.10 and -2.64 respectively, indicating item 2 is an easier item compare to item 1. A high positive value for item 22 (1.25) indicates that this is a difficult item. The difficulty index show the likewise with only 26% of the students obtained a correct answer for this question.

Table 2 Comparison of PROC NL MIXED and PROC GLIMMIX estimated coefficients and Difficulty Index

Question	Difficulty Index	PROC NL MIXED	PROC GLIMMIX	MULTILOG-MG	R Package eRm
1	0.71	-1.14	-1.10	-1.02	-0.93
2	0.91	-2.72	-2.64	-2.44	-2.47
3	0.52	-0.12	-0.11	-0.10	0.09
4	0.53	-0.16	-0.15	-0.14	0.05
5	0.62	-0.63	-0.61	-0.57	-0.43
6	0.53	-0.20	-0.19	-0.17	0.01
7	0.60	-0.55	-0.52	-0.48	-0.34
8	0.53	-0.18	-0.17	-0.15	0.03
9	0.37	0.64	0.62	0.58	0.85
10	0.58	-0.41	-0.40	-0.37	-0.21
11	0.64	-0.74	-0.72	-0.66	-0.54
12	0.67	-0.92	-0.88	-0.82	-0.71
13	0.45	0.21	0.21	0.20	0.42
14	0.59	-0.46	-0.44	-0.41	-0.26
15	0.53	-0.18	-0.17	-0.16	0.02
16	0.37	0.67	0.65	0.61	0.88
17	0.50	-0.05	-0.04	-0.04	0.16
18	0.49	0.03	0.03	0.03	0.24
19	0.56	-0.33	-0.31	-0.29	-0.12
20	0.46	0.18	0.17	0.16	0.38
21	0.64	-0.76	-0.74	-0.68	-0.56
22	0.26	1.29	1.25	1.16	1.51
23	0.48	0.09	0.09	0.08	0.30
24	0.44	0.30	0.29	0.28	0.51
25	0.62	-0.65	-0.63	-0.58	-0.45
26	0.65	-0.82	-0.79	-0.73	-0.61
27	0.34	0.82	0.79	0.74	1.04
28	0.33	0.90	0.87	0.81	1.11

Disadvantages of using PROC NL MIXED

We have seen in the last section that the syntax of PROC NL MIXED is much more complicated compared to PROC GLIMMIX as the mathematical representations need to be explicitly specified in the procedure. On top of this difference, it takes much longer time to run PROC NL MIXED. More seriously, it could possibly end up non-convergence after a long run especially for large sample. For the current sample of 815 students, it takes 7 seconds to estimate the PROC GLIMMIX for the regular Rasch model whereas the PROC NL MIXED takes up to an hour to carry out the same task. From past experience, when the sample size is substantial large, it may take more than a day to run PROC NL MIXED. In times when non convergence occurs during processing, we may have to rely on prior research to specify the starting values or based on classical statistics like difficulty index to provide better starting values for the model to converge. While PROC NL MIXED could usually end up in non-convergence when the sample size is large, PROC GLIMMIX generally does not arise as an issue of concern. However, when there are too many dummy variables specified in the model, PROC GLIMMIX may face the problem as well. This will occur when we specify person as random instead of item as random, the next Rasch model we are going to discuss.

Random Person?

The common practice of Rasch modeling is to treat items as fixed and persons as random. However, random item parameters also make sense theoretically and in practice. If our interest turns from measurement of persons instead of items, we could specify item as random and person as fixed (De Boeck, 2008). Specifying random person distribution is not unexpected as a group of students sampled is generally from population. However, items do not generally exist as a population to be drawn. We do not expect there exists a population of items and the current items used for testing is a sample out of it, even for computer adaptive testing, the items need to be created in the item bank. However, in practice, there exists real life application such as randomly drawn words from a vocabulary. The great advantage of specifying person random is that we could know the distribution of person in order to make comparison of persons who take the test by their estimated coefficients. De Boeck (2008) classifies Rasch models into four categories depending on whether item and/or person are treated as random or fixed. Table 3 below lays out these four types of Rasch models. The regular Rasch is random person fixed item Rasch model.

Table 3 Classification of Rasch Models

Persons	Items	
	Fixed	Random
Fixed	Fixed Person Fixed Item Rasch	Fixed Person Random Item Rasch
Item	Random Person Fixed Item Rasch	Random Person Random Item Rasch

Random Item, Fixed Person Rasch Model

The Random Item, Fixed Person Rasch model is specified as follows as a binary multilevel logistic model.

$$\text{Logit}(\pi_{ij}) = \beta_{oi} + \sum_{k=1}^{J-1} \beta_k Z_{kj}, \quad Y_{ki} \sim \text{Bernoulli}(\pi_{ij}), \quad Z_{kj} = \begin{cases} 1 & \text{if } k = j \\ 0 & \text{otherwise} \end{cases}$$

$$\beta_{oi} = \gamma_{00} + \mathbf{u}_{oi}, \quad \mathbf{u}_i \sim N(\mathbf{0}, \sigma_u^2)$$

Similar to the regular Rasch model, it is a two-level mixed binary logistic model, however now the set of person dummy variables Z_{kj} is specified instead of item dummy variables. The syntax of PROC GLIMMIX for the Random Person Fixed Item Rasch model for the first 50 students is as follows:

```
Proc GLIMMIX Data = LongMathSec350ID;
  Class Item;
  Model LongAccuracy (Event='1') = ID2-ID50
    / Dist = Binary Solution;
  Random Intercept / Subject= Item;
Run;
```

Compared to the Random Person Fixed Item regular Rasch model, there is just a slight change in the PROC GLIMMIX specification. Instead of specifying person as the classification variable, now the item becomes the SUBJECT and the CLASS statement states that it is a classification variable. The independent variables now are the student dummy variables running from ID2 to ID50. It is also noted that student one (ID1) is omitted from the

model statement since there are 50 students, one has to be excluded in model specification. Table 4 shows the estimated coefficients using PROC GLIMMIX of the first 5 students, the number, and the percentage of items correctly answered.

Table 4 First 5 students item profile and PROC GLIMMIX estimated coefficients

Student	No. of Items Correctly Answered	% of Items Correctly Answered	PROC GLIMMIX Estimated Coefficients
1	15	0.54	-0.14
2	14	0.50	-0.32
3	21	0.75	1.01
4	23	0.82	1.52
5	17	0.61	0.21

The second and third columns of Table 4 show the number of items correctly answered, and the classical measurement of the percentage of items correctly answered to assess the students. The estimated coefficients of Random Item Fixed Person Rasch model is in the last column. They give the relatively positioning of students' abilities. Lower ability students have negative coefficients while higher ability students have positive coefficients. For instance, student 4 with the positive coefficient of 1.52 is higher in assessment score compared to student 2 with a negative coefficient of -0.32.

Cross-Classified Model

The above two Rasch models specify either item or person as random and the other as fixed. Nevertheless, both item and person could be specified as random using a cross-classified binary logistic model. Figure 1 below shows the schematic of a cross-classified model for 2 students and 3 items. If both items and persons are regarded as random, we could specify two random residual distributions each for the item and person with responses nested within persons and items. A two-level cross-classified model is specified as shown below.

$$\text{Logit}(\pi_{ij}) = \beta_o + \mathbf{u}_{1j} + \mathbf{u}_{2i}, \quad \mathbf{u}_{1j} \sim N(\mathbf{o}, \sigma_{u1}^2) \text{ and } \mathbf{u}_{2i} \sim N(\mathbf{o}, \sigma_{u2}^2)$$




Figure 1. Schematic of cross-classified with students and items

The PROC GLIMMIX syntax to generate the random item, random person Rasch model is stated below.

```
Proc GLIMMIX Data = LongMathSec3;
  Class Item ID;
  Model LongAccuracy (Event='1') = / Dist = Binary Solution;
  Random Intercept / Subject = Item;
  Random Intercept / Subject = ID;
Run;
```

As both item and person are specified as random, we do not need to specify dummy variables for item and person as both are quantified as distributions. Since there is no covariate in this model, it generates two estimates that partitioned that variance for both the item and student levels. The output of the estimates of cross-classified model is shown below. The student level variance (1.1256) has higher variability than item level variance (0.5415). About two thirds (67.52%) of student performance is at the student level whereas about one-third (32.48 %) is due to item differences.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error
Intercept	Item	0.5415	0.1502
Intercept	ID	1.1256	0.06872

Cross-Classified Model with Covariates

The cross-classified model mentioned above assumes item difficulties and person abilities are independently and normally distributed with mean zero and variance $\sigma_{u_1}^2$ and $\sigma_{u_2}^2$ respectively. This implies that there is no prior assumption that an item is more difficult than another item and a specific person would perform better than another person. These assumptions are often not realistic in real life application. Van den Noortgate et al (2003) mentioned three conditions that could violate these assumptions. Person variables such as age, and grouping could affect ability. Item characteristics such as item domain could also be related to item difficulty, and differential item functioning could also be one of the important factors affecting the probability of students answering a particular item. This leads us to incorporate covariates into the model.

In education research, we would generally like to know whether there are factors at the person and item levels that could affect test results so that we could come out with tests that are fair and without bias. For instance, we may want to know whether the subject domain would have differential effect on performance. We would like to know whether geometry items are differentiated from other mathematical domains. In mathematical testing, the literature has shown that male performed better in geometry and females were better in algebra. These research questions could be carried out using the cross-classified model. When we specify a cross-classified model with covariates, three kinds of covariates could be incorporated namely item covariates, person covariates, and person-by-item covariates. As shown by the general equation below, the three kinds of covariates could be easily incorporated into the cross-classified logistic model by including these covariates in the equation.

$$\text{Logit}(\pi_{ij}) = \beta_o + \sum_{a=1}^A \beta_{ai} \text{Item}_{ai} + \sum_{b=1}^B \beta_{bi} \text{Person}_{bj} + \sum_{c=1}^C \beta_{cij} \text{Item}_{ci} \times \text{Person}_{cj} + u_{1j} + u_{2i}$$

Covariate at person level

The syntax of PROC GLIMMIX to include covariates at person level is shown below. The variable LongConfidence is the confidence level of students in answering the test items. To include this person level covariate, just place it after the “=” sign of the MODEL statement.

```
Proc GLIMMIX Data = LongMathSec3;
  Class Item ID;
  Model LongAccuracy (Event='1') = LongConfidence / Dist = Binary Solution;
  Random Intercept / Subject = Item;
  Random Intercept / Subject= ID;
Run;
```

The output of the model shows that the variable LongConfidence is significant with an estimated coefficient of 2.0973. This means that the higher the confidence of a student of getting a correct answer, the higher the probability the student would score a correct answer. This variable has reduced the student level variance from 1.1256 down to 0.5983, showing a high explanatory power of this variable in explaining student’s performance at the student level.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error
Intercept	Item	0.3972	0.1109
Intercept	ID	0.5983	0.04173

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	-0.9914	0.1269	27	-7.81	<.0001
LongConfidence	2.0973	0.05428	22033	38.64	<.0001

Covariate at item level

Similar to the inclusion of covariates at the person level, covariates at the item level is placed after the “=” sign of the MODEL statement. The following model includes three item level dummy variables in the model. These are dummy variables to denote the mathematical domain of the items. There are altogether four mathematic domains: algebra, geometry, statistics, and number. The reference domain is number. As such, only three dummy variables are specified in the PROC GLIMMIX.

```
Proc GLIMMIX Data = LongMathSec3;
  Class Item ID;
  Model LongAccuracy (Event='1') = Algebra Geometry Statistics
    / Dist = Binary Solution;
  Random Intercept / Subject = Item;
  Random Intercept / Subject= ID;
Run;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error
Intercept	Item	0.4660	0.1372
Intercept	id	1.1263	0.06876

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	0.7915	0.2835	24	2.79	0.0101
Algebra	-0.5037	0.3714	22034	-1.36	0.1751
Geometry	-0.8461	0.3970	22034	-2.13	0.0331
Statistics	-0.08646	0.3711	22034	-0.23	0.8158

The results show that geometry items are marginally significant from the number items, however not for the algebra and statistics domains. The reduction of the variances occurs mainly at the item level. It reduces the variance from 0.5415 of the null model to 0.4660 of the current model.

Covariate at both Person and Item Level

The two above mentioned models examine the person and item covariates separately. It seems obvious that the next step in modeling is to include both the covariates into the model. This is rather straight forward for PROC GLIMMIX. What we have to do is to include both the person and item level covariates after the "=" sign of the MODEL statement and the rest of the syntax remains the same. The results show that after controlling for student's level of confidence, the domain specific dummy variables are not statistically significant at alpha level .05.

```
Proc GLIMMIX Data = LongMathSec3;
  Class Item ID;
  Model LongAccuracy (Event='1') = LongConfidence Algebra Geometry Statistics
    / Dist = Binary Solution;
  Random Intercept / Subject = Item;
  Random Intercept / Subject= ID;
Run;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error
Intercept	Item	0.3825	0.1132
Intercept	id	0.5990	0.04178

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	-0.5971	0.2589	24	-2.31	0.0300
LongConfidence	2.0953	0.05430	22033	38.59	<.0001
Algebra	-0.3216	0.3373	22033	-0.95	0.3403
Geometry	-0.6445	0.3604	22033	-1.79	0.0737
Statistics	0.07410	0.3369	22033	0.22	0.8259

CONCLUSION AND SUGGESTIONS

This paper proposed using PROC GLIMMIX for Rasch modeling as it could be easily specified and takes much shorter time to estimate the model, compared to the PROC NLMIXED. The estimated coefficients of both PROC NLMIXED and PROC GLIMMIX are comparable to standard IRT software MULTILOG-MG, and R package eRm. Four variations of Rasch model were illustrated. Particularly, the Random Item Fixed Person Rasch model gives an alternative perspective in analysis that shifts from item centered to person centered. More importantly, the Random Item Random Person Rasch model gives a new arena in analysis that incorporates covariates into the model. As these Rasch models could be easily carried out using the PROC GLIMMIX, we encourage the use of PROC GLIMMIX for Rasch modeling.

ACKNOWLEDGEMENTS

This paper acknowledges the Centre for Research in Pedagogy and Practice (CRPP), National Institute of Education, Singapore for making the data available from the research project “Core 2 Panel 2 Project” (R59702705A). We would also like to thank Albert Liau, Ridzuan, Gu Fei and Matthias von Davier who took time to go through the paper and give their comment.

REFERENCES

Adams, R. J., Wilson, M., & Wu, M. (1997). Multilevel item response models: An approach to errors in variables regression. *Journal of Educational and Behavioral Statistics*, 22, 47-76.

De Boeck, P., and Leuven, K. U. (2008). Radom Item RASCH Models. *Psychometrika*, 73(4), 533-559.

Kamata, A. (2001). Item analysis by the hierarchical generalized linear model. *Journal of Educational Measurement*, 38, 79-93.

Rasch, G. (1960). *Probabilistic models for some intelligence and attainment tests*. Chicago: Mesa Press.

SAS/STAT® 9.1 User's Guide (2008). The GLIMMIX Procedure. SAS Publishing.

Van den Noortgate, W., De Boeck, P., and Meulders, M. (2003). Cross-classification multilevel logistic models in psychometrics. *Journal of Educational and Behavioral Statistics*, 28, 369-386.

AUTHOR CONTACT INFORMATION

Teck Kiang Tan
Centre for Research in Pedagogy and Practice
Nanyang Technological University
1 Nanyang Walk, Block 5, Basement 3
Singapore 637616
Work Phone: 065-62196277
Email: teckkiang.tan@nie.edu.sg

Serena Luo Wen Su
Centre for Research in Pedagogy and Practice
Nanyang Technological University
1 Nanyang Walk, Block 5, Basement 3
Singapore 637616
Work Phone: 065-62196277
Email: wensu.luo@nie.edu.sg

David Hogan
Centre for Research in Pedagogy and Practice
Nanyang Technological University
1 Nanyang Walk, Block 5, Level 3
Singapore 637616
Work Phone: 065-62196277
Email: david.hogan@nie.edu.sg

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.